

工作日志

| | | | |
|--------|--|----|------------|
| 标题 | 0910 工作日志 | | |
| 姓名 | 汪能志 | 学号 | U201713082 |
| 日期 | 2020年9月 10 日 | 天气 | 雨 |
| 日记撰写时间 | | | |
| 内容 | <p>一、0909 实验问题</p> <p>在昨天的闪灯实验中，进行循环的三个变量所处的位置并不是 common ram，在循环过程中也没有进行 ram bank 的切换设置，这可能会导致内存地址错误的问题。这个问题有两种解决方法，一是将变量的地址设置为 common ram，这样在不同的 bank 中所见的值都相同，也就不会出现定位错误的问题；二是使用可重定位的编程，让连接器来分配空间。</p> <p>二、实验目的</p> <p>利用 timer0 定时器和定时器查询的方式，实现周期为 1s 的闪灯。</p> <p>三、定时器原理</p> <p>PIC16F18854 的 timer0 定时器的结构如下图：</p> <p>对于 8-bit 定时器来说，输入的信号在经过分频后，驱动 8 位循环计数器对寄存器 TMR0L 进行累加计数。在 TMR0L 和预置的 TMR0H 相等是，就会触发一次高电平。</p> <p>通过在主循环中查询溢出位的值，就可以实现在周期性地闪灯。</p> <p>根据系统架构图，prescaler 和 postscaler 的分频系数的乘积应该为实际的等效分频系数。timer 的输出时钟周期间隔约为：$prescaler \times postscaler \times (TMR0H + 1)$</p> <p>在本次实验中，系统时钟为 4Mhz，因此输入定时器的内部指令时钟的周期为 1us。为了达到定时闪灯的效果，需要 timer0 能输入约 500000 个周期的时钟信号后，产生一个脉冲信号。</p> <p>由于 8-bit 定时器能实现的最长脉冲周期约为 $2^{15} \times 16 \times 256 = 2^{27}$，远超实验所需，因此本次实验中我们小组使用的是 8-bit 定时器来完成实验。</p> <p>四、定时器实现</p> <pre> ; init TMR0 BANKSEL TMR0H MOVLW 0xF3 MOVWF TMR0H </pre> | | |

```
BANKSEL    TMR0L
MOVLW     0x00
MOVWF     TMR0L

BANKSEL    T0CON0
MOVLW     B'11001111';
; bit7     enable timer0
; bit4     8-bit timer
; bit 3-0   postscaler 1:16
MOVWF     T0CON0

BANKSEL    T0CON1
MOVLW     B'01000111';
; bit 7-5   clk source F_OSC / 4
; bit 4     sync
; bit 3-0   prescaler 1:128
MOVWF     T0CON1

; clear the output signal
BANKSEL    PIR0
BCF        PIR0, 5
; =====

LOOP
    BANKSEL    PIR0
    BTFSS     PIR0, 5
    ; no output from TMR0
    GOTO      LOOP
    ; have output from TMR0
    BANKSEL    PIR0
    BCF        PIR0, 5          ; clear TMR0IF

    BANKSEL    TMR0L
    MOVLW     0x00
    MOVWF     TMR0L

    BANKSEL    PORTC
    MOVLW     B'11111111'
    XORWF     PORTC, f          ; flip PORTC
    GOTO      LOOP
END
```

以上设置的定时器的时间据计算为： $128 \times 16 \times (243 + 1) + 9 = 499721$ ，需要注意，查询循环中的内容会引入一些多余的指令数。

```
Target halted. Stopwatch cycle count = 499721 (499.721 ms)
Target halted. Stopwatch cycle count = 499721 (499.721 ms)
Target halted. Stopwatch cycle count = 499721 (499.721 ms)
```

通过跑表进行调试，和调试结果相同。

五、实验结果

将代码导入开发板，开发板可以按大约 1s 的周期闪烁。和预期结果相同。

六、实验总结

本次实验相比上次实验更为复杂，在编写过程中也遇到了更多的问题，例如在网上查找到的相关资料主要都是基于 PIC16F887 单片机，其定时器相关的内容和实验课所使用的单片机有较大的区别，因此对于实际操作的部分都需要重新编写。由于汇编语言是直接面向硬件进行编程，在编写过程中需要时刻回归单片机的硬件架构设计和相关的说明文档进行编写和调试。